DECUS
PROGRAM LIBRARY

DECUS NO.	8-521
TITLE	A CLOCK
AUTHOR	Klaus Lickteig
COMPANY	Technische Universitaet Berlin Berlin, Germany
DATE	March 1972
SOURCE LANGUAGE	PAL III

A CLOCK

DECUS Program Library Write-up

DECUS NO. 8-521

ABSTRACT:

This demonstration program will display a clock on the oscylloscope of an AXØ8 A/D-converter. After setting the clock, the running clock will displayed on the oscylloscope. There are possibilities to regulate the clock during running.

REQUIREMENTS:

Storage: This program occupies the locations Ø...2, 1Ø, 2Ø...57, 2ØØ...1145.
The data are in locations 1146...4347

Equipment: PDP-8/I or PDP-8/E with an AXØ8 A/D-converter and ASR-33 teletype (LAB-8 System)

LOADING PROCEDURE:

Load the binary tape into field Ø with the BIN-Loader.
Turn on the teletype and the oscylloscope of AXØ8.

WORKING OF PROGRAM:

1. After loading, start the program at location Ø2ØØ.
   The teletype will type a carriage-return, the oscylloscope will display the face of the clock,
   the hands of the clock will be at 12.

2. Type the hours on the keyboard-reader
   (ØØ, Ø1,...,11). If the input is correct, it will
   be typed on the teletype.

3. Type the minutes on the keyboard-reader (ØØ, Ø1,..., 59).
   If the input is correct, it will be typed on the teletype.
   During this operations the clock face will be displayed
   on the oscylloscope, it will show the time.

4. If you type any sign on the keyboard-reader, the
   clock will run (controlled by the crystal-clock
   of the AXØ8) and display the correct time on the
   oscylloscope.

5. While the clock is running there are two
   possibilities of input:

M	minute hand one minute foward, second hand to 12
S	second hand to 12

/ KLAUS LICKTEIG
/ INSTITUT FUER KERNTECHNIK
/ TECHNISCHE UNIVERSITAET BERLIN
/ MARCHSTRASSE 18
/ 1000 BERLIN 10
/ GERMANY

/ PROGRAM: DISPLAY A CLOCK ON THE OSCYLLOSCOPE

/ EQUIPMENT: PDP-8
/ AX08
/ OSCYLLOSCOPE

/ STORAGE: FOR PROGRAM: 1,2,10,20...57,200...1145
/ FOR DATA: 1146...4347

/ LOADING PROCEDURE
/ LOAD THE BINARY-TAPE AND START THE PROGRAM AT LOCATION 0200.
/ THE TELETYPE WILL TYPE A CARRIAGE-RETURN, THE OSCYLLOSCOPE
/ WILL DISPLAY THE FACE OF THE CLOCK, THE HANDS OF THE CLOCK
/ WILL BE AT 12.

/ WORKING OF PROGRAM
/ TYPE THE HOURS ON KEYBOARD-READER (00,01,...,11), IF THE
/ INPUT IS CORRECT, IT WILL BE TYPED ON THE TELETYPE. THEN
/ TYPE THE MINUTES ON KEYBOARD-READER (00,01,...,59), IF THE
/ INPUT IS CORRECT IT WILL BE TYPED ON THE TELETYPE. DURING
/ THE INPUT THE CLOCK FACE WILL BE DISPLAYED ON THE OSCYLLOSCOPE
/ IT WILL SHOW THE TIME.
/ AFTER THE INPUT OF HOURS AND MINUTES THE CLOCK WILL RUN
/ (CONTROLLED BY THE CRYSTAL-CLOCK OF THE AX08), IF YOU TYPE
/ ANY SIGN ON THE KEYBOARD.
/ DURING RUNNING THERE ARE TWO POSSIBILITIES OF INPUT:
/ M ONE MINUTE FORWARD AND SECONDS TO 12
/ S SECONDS TO 12

```
/ SPECIFICATION FOR THE AX08

CLXK=6352
SKXK=6321
OTEN=6344
DXC=6301
DXL=6302
DYC=6311
DYL=6312
DIS=6314
```



```
*1
        JMP I .+1           / POINTER TO THE
        INTER               /     / INTERRUPT ROUTINE



*20

START,  IOUT                / IOUT+1 IS START OF DATA
POINTER, 0                  / INEW+480
M60,    -74                 / -60
C74,    112                 / 74
CNT,    -3420               / -10.000
        -3
M14,    -14                 / -12
SECOND, 0
MINUTE, 0
HOURE,  0
DISS,   0
DISM,   0
DISH,   0
LOW,    0
HIGH,   0
MIN60,  0
M15,    -17                 / -15
        -20                 / -16
SEC15,  0
MIN15,  0
HOU15,  0
X,      0
Y,      0
QUADSEC, 0
XS,     0
YS,     0
QUADMIN, 0
XM,     0
YM,     0
QUADH,  0
XH,     0
YH,     0
```

```
/ PROGRAM-START AT START ADRESS 0200

*200

          JMP I .+1              / POINTER TO THE
          GO                     /     / PROGRAM-START



/ INTERRUPT-ROUTINE

INTER,    SKXK                   / INTERRUPT FROM CRYSTAL-CLOCK ?
          JMP NOCRYS             / NO: FROM TTY-INPUT

          CLXK                   / YES: CLEAR CRYSTAL-CLOCK FLAG
          ISZ LOW
          JMP GOON20
          ISZ HIGH               / 10.000-TIMES INTERRUPT ?
          JMP GOON20             / NO

          DCA SAV
          TAD CHECK
          SNA CLA                / READY ?
          HLT                    / NO: ERROR
          TAD SAV                / YES
          JMS SAVE               / SAVE (AC),(LINK),(0)
          JMS NEW                / SET COUNTER (-10.000), ION

          ISZ SECOND             / SEC=SEC+1-->60=0 ?
          SKP                    / NO
          JMP GOON6              / YES: CHECK MINUTE
          JMS I .+1              / NO: SET A NEW START OF
          NEWSEC                 /     / THE SECONDS HAND
          JMP GOON9              /     / OF THE CLOCK

GOON6,    JMS ZEROSEC            / YES: 60 SECONDS
          ISZ MIN60              / 60 MINUTES ?
          JMP GOON8              / NO

          TAD M60                / YES: 60 MINUTES
          DCA MIN60              / SET A NEW COUNTER
          TAD M15+1
          DCA MIN15
          DCA QUADMIN            / SET 1-ST QUARTER
          DCA XM
          DCA YM
          TAD M14
          DCA MINUTE
          TAD POINTER            / NEW START OF THE MINUTES
          DCA DISM               /     / HAND OF THE CLOCK
          ISZ HOURE              / HOURE=HOURE+1-->60=0 ?
          JMP GOON7              / NO
```

```
            TAD M60             / YES: SET A  NEW COUNTER
            DCA HOURE
            TAD M15+1
            DCA HOU15
            DCA QUADH           / SET 1-ST QUARTER
            DCA XH
            DCA YH
            TAD POINTER         / NEW START OF THE HOURS
            DCA DISH            /     / HAND OF THE CLOCK
            JMP GOON9

GOON7,      JMS I .+1           / SET START OF THE
            NEWHOU              /     / HOURS HAND OF
            JMP GOON9           /     / THE CLOCK

GOON8,      JMS I .+1           / NEW START OF THE MINUTES
            NEWMIN              /     / HAND OF THE CLOCK
            ISZ MINUTE          / 12 MINUTES = 1/5 HOURE ?
            JMP GOON9           / NO

            TAD M14             / YES: SET NEW COUNTER
            DCA MINUTE
            JMS I .+1           / NEW START OF THE HOURS
            NEWHOU              /     / HAND OF THE CLOCK
            ISZ HOURE

GOON9,      CLA CLL
            IOF                 / INTERRUPT OFF
            CMA
            DCA CHECK           / SET CHECK-LOCATION TO -1
            TAD ZERO
            DCA Ø               / FETCH OLD (Ø)
            TAD SAVEL
            RAL                 / FETCH OLD (LINK)
            TAD SAVE            / FETCH OLD (AC)

GOON2Ø,     ION                 / INTERRUPT ON
            JMP I Ø             / RETURN: END OF ROUTINE

NOCRYS,     DCA SAV             / SAVE (AC)
            TAD CHECK
            SZA CLA             / READY FOR INPUT ?
            JMP .+5             / YES

            KRB                 / NO: CLEAR FLAG AND BUFFER
            CLA CLL             / IGNORE INPUT
            TAD SAV             / FETCH (AC)
            JMP GOON2Ø          / RETURN
```

```
            TAD SAV             / YES: READY FOR INPUT
            JMS SAVE            / SAVE (AC),(LINK),(Ø)
            ION                 / INTERRUPT ON
            KRB
            TAD M315
            SZA                 / WAS INPUT "M" ?
            JMP GOON1Ø          / NO

            JMS NEW             / YES: SET THE CLOCK ONE
            JMP GOON6           / MINUTE FOWARD

GOON1Ø,     TAD M6
            SZA CLA             / WAS INPUT "S" ?
            JMP GOON9           / NO

            JMS NEW             / YES
            JMS ZEROSEC         / ZERO SECONDS
            JMP GOON9



/ SUBROUTINE TO SET THE SECONDS HAND OF THE CLOCK TO ZERO

ZEROSEC,    HLT
            TAD M6Ø             / SET NEW COUNTER (-6Ø)
            DCA SECOND
            TAD M15+1           / (-16)
            DCA SEC15
            DCA QUADSEC         / SET 1-ST QUARTER
            DCA XS
            DCA YS
            TAD POINTER         / SET NEW START OF SECONDS
            DCA DISS            /     / HAND OF THE CLOCK
            JMP I ZEROSEC       / RETURN: END OF SUBROUTINE



/ SUBROUTINE TO SAVE THE CONTENTS OF (AC),(LINK),(Ø)
/       AND TO SET THE CHECK-LOCATION TO ZERO

SAVE,       HLT
            DCA SAVEAC          / SAVE (AC)
            RAR
            DCA SAVEL           / SAVE (LINK)
            TAD Ø
            DCA ZERO            / SAVE (Ø)
            DCA CHECK           / CHECK-LOCATION=Ø
            JMP I  SAVE         / RETURN: END OF SUBROUTINE
```

```
/ SUBROUTINE TO SET THE COUNTER (-10.000) AND
/       TO TURN THE INTERRUPT ON

NEW,        HLT
            CLA CLL
            TAD CNT             / SET COUNTER (-10.000)
            DCA LOW
            TAD CNT+1
            DCA HIGH
            ION                 / INTERRUPT ON
            JMP I NEW           / RETURN: END OF SUBROUTINE



/ SYMBOLS

M6,         -6
M315,       -315
SAVEAC,     0
SAVEL,      0
SAV,        0
ZERO,       0
CHECK,      0
```



```
/ SYMBOLS

DIES,       DISPL
IN,         INPUT
M12,        -12
CAR,        CARRET
CNT1,       0
C2,         2
C480,       740             / 480
C60,        74              / 60
C14,        14              / 12
IGOON5,     GOON5
```

```
/ START OF MAIN-PROGRAM

GO,     KCC                     / CLEAR FLAGS
        TLS                     / AND BUFFER
        RRB
        PCF
        CLA CLL
        JMS I CAR               / TYPE A CARRIAGE-RETURN

        TAD M15+1               / SET COUNTER (-16)
        DCA MIN15
        TAD M15+1
        DCA HOU15
        DCA QUADMIN             / SET 1-ST QUARTER
        DCA XM
        DCA YM
        DCA QUADH
        DCA XH
        DCA YH
        TAD START               / START OF CLOCK-FACE
        TAD C480
        DCA POINTER             / START OF THE HANDS
        TAD POINTER             /     / OF THE CLOCK
        DCA DISS                /     SECOND = 12
        TAD POINTER
        DCA DISM                /     MINUTE = 12
        TAD POINTER
        DCA DISH                /     HOURE = 12
        JMS I DIES              / DISPLAY ON OSC., INPUT ?

        JMS I IN                / YES: 1-ST INPUT OF HOURS
        -2                      / ONLY 0,1
        SPA SNA                 / WAS INPUT ZERO ?
        JMP .+4                 / YES
        CLL RAL                 / NO
        CLL CML RTL
        SKP
        CLA CLL
        DCA HOURE
        TAD HOURE
        SZA CLA                 / WAS IT ZERO ?
        JMP .+3                 / NO: INPUT WAS 1
        TAD M12                 / YES: INPUT WAS 0
        SKP
        CLA CLL CMA RAL
        DCA .+3                 / SET CHECK FOR 2-ND INPUT

        JMS I DIES              / DISPLAY ON OSC., INPUT ?
```

```
        JMS I IN            / YES: 2-ND INPUT OF HOURS
        Ø                   / ONLY Ø,1 OR Ø,1,...,9
        TAD HOURE
        DCA HOURE
        TAD HOURE           / GET NUMBER OF HOURS
        CLL RTL             /     / AND MULTIPLY IT BY 5
        TAD HOURE
        DCA HOURE           / NUMBER OF 1/5 HOURS

        TAD HOURE           / SET COUNTER OF HOURS
        TAD M6Ø
        DCA HOURE
        TAD C6Ø
        TAD HOURE
        SNA CLA             / COUNTER OF HOURS (-6Ø) ?
        JMP GOON1           / YES
        TAD C6Ø             / NO: SET START OF THE
        TAD HOURE           /     / HOURS HAND OF
        CIA                 /     / THE CLOCK
        DCA CNT1
        JMS I .+1
        NEWHOU
        ISZ CNT1
        JMP .-3

GOON1,  JMS I CAR           / TYPE A CARRIAGE-RETURN
        JMS I DIES          / DISPLAY ON OSC., INPUT ?

        JMS I IN            / YES: 1-ST INPUT OF MINUTES
        -6                  / ONLY Ø,1,...,5
        SNA                 / WAS INPUT ZERO ?
        JMP GOON2           / YES
        DCA MINUTE          / NO
        TAD MINUTE
        CIA
        DCA CNT1            / SET COUNTER
        TAD C2              / DECIMAL TO OCTAL CONVERSION
        ISZ CNT1
        JMP .-2
        DCA CNT1
        TAD MINUTE
        CLL RAL
        RTL
        TAD CNT1
GOON2,  DCA MINUTE          / SET 1-ST INPUT OF MINUTES

        JMS I DIES          / DISPLAY ON OSC., INPUT ?
```

```
          JMS I IN            / YES: 2-ND INPUT OF MINUTES
          -12                 / ONLY 0,1,...,9
          TAD MINUTE
          DCA MINUTE
          JMS I CAR           / TYPE A CARRIAGE-RETURN
          TAD MINUTE
          TAD M60
          DCA MIN60           / SET COUNTER OF MINUTES
          TAD MINUTE          / CORRECT THE HOURS HAND
          DCA CNT1            /     / OF THE CLOCK
GOON3,    TAD CNT1
          TAD M14
          SPA                 / ARE MINUTES > 12 ?
          JMP GOON4           / NO
          DCA CNT1            / YES
          JMS I .+1           / CORRECT START OF THE
          NEWHOU              /     / HOURS HAND OF CLOCK
          ISZ HOURE           / CORRECT COUNTER OF HOURS
          TAD M14
          DCA XS
          JMS I .+1           / SET START OF THE
          NEWMIN              /     / MINUTES HAND OF
          ISZ XS              /     / THE CLOCK
          JMP .-3
          JMP GOON3

GOON4,    DCA MINUTE          / SET COUNTER OF MINUTES
          TAD C14
          TAD MINUTE
          SNA CLA             / COUNTER OF MINUTES (-12) ?
          JMP I IGOON5        / YES
          TAD C14             / NO
          TAD MINUTE
          CIA                                          Page 3
          DCA XS
          JMS I .+1           / CORRECT START OF THE
          NEWMIN              /     / MINUTES HAND OF
          ISZ XS              /     / THE CLOCK
          JMP .-3

GOON5,    TAD MM60
          DCA SECONDS         / SET COUNTER OF SECONDS (-60)
          TAD M15+1           / SET COUNTER (-16)
          DCA SEC15
          TAD QUADSEC         / SET 1-ST QUARTER
          DCA XS
          DCA YS
```

```
TSF
JMP .-1
TCF                     / CLEAR FLAG
JMS DISPL               / DISPLAY ON OSC., INPUT ?
KRB                     / YES
CLA CLL CMA
DCA I CHECKI
CLXK
SKXK                    / SYNCHRONIZE THE
JMP .-1                 /     / CRYSTAL-CLOCK
CLXK
JMS I INEW              / SET COUNTER (-10.000); ION
TAD C400
OTEN                    / INTERRUPT FROM CRYSTAL-CLOCK

JMS DISPLAY             / DISPLAY THE CLOCK
JMP .-1                 /     / ON OSCYLLOSCOPE
```

```
/ SUBROUTINE FOR AN INPUT OF A NUMBER. IF IT IS CORRECT,
/       TYPE IT ON TTY

INPUT,      HLT
            KRB                     / FETCH INPUT
            DCA NUMBER
            TAD NUMBER
            TAD M260
            SPA
            JMP AGAIN               / INPUT AGAIN
            TAD I INPUT
            SMA CLA                 / WAS INPUT CORRECT ?
            JMP AGAIN               / NO: INPUT AGAIN
            TAD NUMBER              / YES
            JMS OUTPUT              / PRINT IT ON TTY
            TAD NUMBER
            AND M0017
            ISZ INPUT
            JMP I INPUT             / RETURN

AGAIN,      CLA CLL CMA RAL         / INPUT AGAIN
            TAD INPUT
            DCA INPUT
            JMP I INPUT             / RETURN: END OF SUBROUTINE



/ SUBROUTINE FOR OUTPUT ON TTY

OUTPUT,     HLT
            TSF                     / READY ?
            JMP .-1
            TLS                     / YES: PRINT IT
            CLA CLL
            JMP I OUTPUT            / RETURN: END OF SUBROUTINE



/ SUBROUTINE TO DISPLAY THE CLOCK ON OSCYLLOSCOPE
/       AND WAIT FOR INPUT FROM TTY

DISPL,      HLT
            JMS DISPLAY             / DISPLAY ON OSC.
            KSF                     / ANY INPUT ?
            JMP .-2                 / NO
            JMP I DISPL             / YES: RETURN: END OF SUBROUTINE
```

```
/ SUBROUTINE TO DISPLAY THE CLOCK ON OSCYLLOSCOPE

DISPLAY,  HLT
          TAD M240
          DCA CNT10
          DCA X
          DCA Y
          TAD START
          JMS OSC              / DISPLAY THE CLOCK-FACE
          TAD M37
          DCA CNT10
          TAD XS
          DCA X
          TAD YS
          DCA Y
          TAD DISS
          JMS OSC              / DISPLAY THE SECONDS HAND
          TAD M34              /      / OF THE CLOCK
          DCA CNT10
          TAD XM
          DCA X
          TAD YM
          DCA Y
          TAD DISM
          JMS OSC              / DISPLAY THE MINUTES HAND
          TAD M25              /      / OF THE CLOCK
          DCA CNT10
          TAD XH
          DCA X
          TAD YH
          DCA Y
          TAD DISH             / DISPLAY THE HOURS HAND
          JMS OSC              /      / OF THE CLOCK
          JMP I DISPLAY        / RETURN: END OF SUBROUTINE


/ SUBROUTINE TO DISPLAY POINTS ON OSCYLLOSCOPE

OSC,      HLT
          DCA 10               / SET AUTO-INDEX REGISTER
          TAD X
          JMS QUAD             / WICH PART ?
          DXC DXL
          CLA CLL
          TAD Y
          JMS QUAD             / WICH PART ?
          DYC DYL DIS          / DISPLAY THE POINT
          CLA CLL
          ISZ CNT10
          JMP OSC+2
          JMP I OSC            / RETURN: END OF SUBROUTINE
```

```
/ SUBROUTINE TO PREPARE DATAS FOR DISPLAY

QUAD,     HLT
          SNA CLA
          JMP .+5
          TAD I 10
          CIA
          TAD CIRCLE          / X <-- 2*R-X
          JMP I QUAD          / RETURN
          TAD I 10            / X <-- X
          JMP I QUAD          / RETURN, END OF SUBROUTINE


/ SYMBOLS

NUMBER,   0
M260,     -260
M0017,    17
C400,     400
M240,     -360                / -240
MM60,     -74                 / -60
CNT10,    0
CIRCLE,   233                 / 155
M37,      -45                 / -37
M34,      -42                 / -34
M25,      -31                 / -25
INEW,     NEW
CHECKI,   CHECK
          0
```

```
/ SUBROUTINE TO SET THE SECONDS HAND OF THE CLOCK

NEWSEC,   HLT
          ISZ SEC15          / ARE 15 SECONDS ?
          JMP .+6            / NO

          TAD M15            / YES: SET COUNTER (-15)
          DCA SEC15
          ISZ QUADSEC        / SET NEW QUARTER
          JMS QUADP          /     / AND SET LOCATION
          QUADSEC            /     / XS AND YS

          JMS QUADIS         / SET THE START OF THE
          QUADSEC            /     / SECONDS HAND OF
          DISS               /     / THE CLOCK
          ISZ NEWSEC
          JMP I NEWSEC       / RETURN, END OF SUBROUTINE


/ SUBROUTINE TO SET THE MINUTES HAND OF THE CLOCK

NEWMIN,   HLT
          ISZ MIN15          / 15 MINUTES ?
          JMP .+6            / NO

          TAD M15            / YES: SET COUNTER (-15)
          DCA MIN15
          ISZ QUADMIN        / SET THE NEW QUARTER
          JMS QUADP          /     / AND SET LOCATION
          QUADMIN            /     / XM AND YM

          JMS QUADIS         / SET THE START OF THE
          QUADMIN            /     / MINUTES HAND OF
          DISM               /     / THE CLOCK
          ISZ NEWMIN
          JMP I NEWMIN       / RETURN, END OF SUBROUTINE
```

```
/ SUBROUTINE TO SET THE HOURS HAND OF THE CLOCK

NEWHOU,   HLT
          ISZ HOU15           / ARE 3 HOURS ?
          JMP .+6             / NO

          TAD M15             / YES
          DCA HOU15           / SET COUNTER (-15)
          ISZ QUADH           / SET THE NEW QUARTER
          JMP QUADP           /     / AND SET THE LOCATION
          QUADH               /     / XH AND YH

          JMP QUADIS          / SET THE START OF THE
          QUADH               /     / HOURS HAND OF
          DISH                /     / THE CLOCK
          ISZ NEWHOU
          JMP I NEWHOU        / RETURN, END OF SUBROUTINE



/ SUBROUTINE TO SET THE NEW QUARTER OF THE CLOCK

QUADP,    HLT
          TAD I QUADP
          ISZ QUADP
          DCA QUA1            / FETCH SEC, MIN OR HOURS
          TAD I QUA1          / SET THE WAY OF
          TAD .+4             /    / THE SUBROUTINE
          DCA .+4
          ISZ QUA1
          SKP
          JMP .+1
          0000
          JMP QUAD2           / 2-ND QUARTER
          JMP QUAD3           / 3-RD    "
          JMP QUAD4           / 4-TH    "
QUAD2,    DCA I QUA1          / X <-- X
          ISZ QUA1
          CLA CMA
          DCA I QUA1          / Y <-- 2*R-Y
          JMP I QUADP
QUAD3,    CLA CMA
          DCA I QUA1          / X <-- 2*R-X
          ISZ QUA1
          CLA CMA
          DCA I QUA1          / Y <-- 2*R-Y
          JMP I QUADP
QUAD4,    CLA CMA
          DCA I QUA1          / X <-- 2*R-X
          ISZ QUA1
          DCA I QUA1          / Y <-- Y
          JMP I QUADP         / RETURN, END OF SUBROUTINE
```

```
/ SUBROUTINE TO SET THE START OF THE HAND OF THE CLOCK

QUADIS,   HLT
          TAD I QUADIS          / FETCH THE QUARTER
          ISZ QUADIS
          DCA QUA1
          TAD I QUADIS          / FETCH THE START OF THE
          ISZ QUADIS            /     / HAND OF THE CLOCK
          DCA QUA2
          TAD I QUA1            / SET THE WAY OF SUBROUTINE
          TAD .+5
          DCA .+5
          ISZ QUA1
          TAD C74
          SKP
          JMP .+2
          0000
          JMP QUADI             / 1-ST QUARTER
          JMP QUADI-1           / 2-ND    "
          JMP QUADI             / 3-RD    "
          CIA                   / 4-TH    "
QUADI,    TAD I QUA2
          DCA I QUA2
          JMP I QUADIS          / RETURN, END OF SUBROUTINE



/ SUBROUTINE TO TYPE A CARRIAGE-RETURN ON TTY

CARRET,   HLT
          TAD C215
          JMS I IOUT
          TAD C212
          JMS I IOUT
          JMP I CARRET          / RETURN: END OF SUBROUTINE



/ SYMBOLS

QUA1,     0
QUA2,     0
C212,     212
C215,     215
IOUT,     OUTPUT



/ START OF DATA BUFFER
```

The data buffer contains the data of the x- and y-coordinates for the oscylloscope of the AXØ8 A/D-converter, i. e. the coordinates for both the face and the hands of the clock. The coordinates are those of the hands of the clock showing 60, 1, 2, ..., 15 seconds. The coordinates of the hands showing 16, 17, ..., 59 seconds will be calculated by the program from the above data.